<commit_before>{ stdenv, src, pkgconfig, tcl, libXft, fontconfig, patches ? [], ... }:

stdenv.mkDerivation {
  name = "tk-${tcl.version}";

  inherit src patches;

  postInstall = ''
    ln -s $out/bin/wish* $out/bin/wish
  '';

  preConfigure = ''
    cd unix
  '';

  configureFlags = [
    "--with-tcl=${tcl}/lib"
  ];

  buildInputs = [ pkgconfig tcl libXft ]
    ++ stdenv.lib.optional stdenv.isDarwin fontconfig;

  NIX_CFLAGS_LINK = if stdenv.isDarwin then "-lfontconfig" else null;

  inherit tcl;

  passthru = rec {
    inherit (tcl) release version;
    libPrefix = "tk${tcl.release}";
    libdir = "lib/${libPrefix}";
  };

  meta = with stdenv.lib; {
    description = "A widget toolkit that provides a library of basic elements for building a GUI in many different programming languages";
    homepage = http://www.tcl.tk/;
    license = licenses.tcltk;
    platforms = platforms.all;
    maintainers = with maintainers; [ lovek323 wkennington ];
  };
}
<commit_msg>tk: propagate dependency on tcl & libXft<commit_after>{ stdenv, src, pkgconfig, tcl, libXft, fontconfig, patches ? [], ... }:

stdenv.mkDerivation {
  name = "tk-${tcl.version}";

  inherit src patches;

  postInstall = ''
    ln -s $out/bin/wish* $out/bin/wish
  '';

  preConfigure = ''
    cd unix
  '';

  configureFlags = [
    "--with-tcl=${tcl}/lib"
  ];

  buildInputs = [ pkgconfig ]
    ++ stdenv.lib.optional stdenv.isDarwin fontconfig;

  propagatedBuildInputs = [ tcl libXft ];

  NIX_CFLAGS_LINK = if stdenv.isDarwin then "-lfontconfig" else null;

  inherit tcl;

  passthru = rec {
    inherit (tcl) release version;
    libPrefix = "tk${tcl.release}";
    libdir = "lib/${libPrefix}";
  };

  meta = with stdenv.lib; {
    description = "A widget toolkit that provides a library of basic elements for building a GUI in many different programming languages";
    homepage = http://www.tcl.tk/;
    license = licenses.tcltk;
    platforms = platforms.all;
    maintainers = with maintainers; [ lovek323 wkennington ];
  };
}

{'fraction_non_alphanumeric': 0.1355764473033152, 'fraction_numerical': 0.0029688273132112814, 'mean_word_length': 3.126530612244898, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '964488', 'n_tokens_mistral': 736, 'n_tokens_neox': 708, 'n_words': 174}